

Department of Electronics and Telecommunication Engineering
University of Moratuwa



EN3030 - ELECTRONIC CIRCUITS AND SYSTEM DESIGN

PROCESSOR DESIGN REPORT

Group-24

S.Prarththanan 180489E
Nerththiga.N 180418M
T.Mayooran 180391V
S.Vithurabiman 180588G

Contents

1	Introduction	4
2	Algorithm	4
3	Instruction Set Architecture	6
3.1	Datapath	7
3.2	Instruction Set	8
3.3	Control Unit	9
3.4	ALU	9
3.5	Mux for B bus	10
3.6	Assembly code of the Algorithms	10
3.7	Instruction Cycle	13
3.8	State Diagram	16
4	RTL Modules	17
4.1	Processor.	17
4.1.1	General Purpose Register	18
4.1.2	Accumulator	18
4.1.3	Memory Address Register (MAR)	18
4.1.4	Program Counter (PC)	19
4.1.5	Memory Buffer Register Unit (MBRU)	19
4.1.6	Instruction ROM (IRAM)	20
4.1.7	Control Unit	20
4.1.8	Decoder	20
4.1.9	Overview of the processor	21
4.2	Data RAM (DRAM)	21
4.3	Clock Generator	22
4.4	Multiplexers and Demultiplexers	22
4.5	Over view of the CPU	24
5	Results and Error analysis	24
6	Appendix	27
6.1	Python Codes	27
6.1.1	Image Downsampling	27
6.1.2	Generating Input Text	28
6.1.3	Generating Image From Text	28
6.1.4	Error Analysis	28
6.2	Verilog Codes	29
6.2.1	ALU	29
6.2.2	Control Unit	30
6.2.3	IRAM	32
6.2.4	Clock Generator	35
6.2.5	Decoder	36
6.2.6	Demultiplexer	36
6.2.7	GPR	37
6.2.8	MAR	37
6.2.9	MBRU	37
6.2.10	MDR	38
6.2.11	Multiplexer (1-bit)	38
6.2.12	Multiplexer (8-bit)	39
6.2.13	Multiplexer (16-bit)	39
6.2.14	PC	39

6.2.15	Processor	40
6.2.16	Downsampling CPU	43

Abstract

FPGAs belong to a class of devices known as programmable logic, or sometimes referred to as programmable hardware. In this project, we down-sample a $256 * 256$ image by a factor of 2. The input of for the FPGA is given as a text file while the output is retrieved as a text file too. To compare the outcome we analyze the FPGA down-sampled image to a python down-sampled image. Verilog is used as the hardware language (HDL). The designed processor and its Instruction Set Architecture is clearly discussed.

keywords : Processor Design, FPGA, Verilog, Down-sampling, ISA

1 Introduction

Objective of this project is to design a custom processor using FPGA, which could accept an image of size 256 x 256 as the input and down sample it to the size of 128 x 128. This report consists of the Algorithm we developed to perform down sampling, Instruction Set Architecture of our processor and other resources.

A central processing unit is an integrated electronic circuit that performs arithmetical, logical, input/output, and other instructions which are essential for the functionality of a computer. Microprocessor is a single chip that includes the functionality of CPU and some few new circuits integrated within. Design of a microprocessor occupies an important part to perform the required task successfully.

2 Algorithm

As we are required to down sample an image, it simply means selecting a few pixels which could represent the original image. So before picking up those selective pixels it is very important to filter the image in order to avoid the aliasing effect. Here we chose 2D Gaussian kernel to smoothen out the input image. So, we could pick the pixels for the down sampled image after filtering.

1	2	1
2	4	2
1	2	1

The 2D convolution of image (I) and Gaussssian kernel(G) can be computed by the below equation.

$$I(x, y) * G(x, y) = \sum_{n_1=-\infty}^{\infty} \sum_{n_2=-\infty}^{\infty} I[n_1, n_1].G[x - n_1, y - n_2]$$

Picking up the pixels for downsampled image can be done as shown below. We would be able to produce a down sampled image of size 3 x 3 from an input image of size 8 x 8.




Figure 1: Downsampling Image

Further, image filtering and downsampling can be integrated together. This could be achieved by smoothening only the selective pixels that are to be used for downsampled image rather than smoothening each and every pixel of the input image. This would help to attain a less complex and comparatively fast processor through avoiding unwanted calculations.

As the first step, the input image should be converted into a text file containing its pixel values in a 1D array. This text file would be used to store the pixel values in data memory. Following that, the down sampling process would be performed by processor and finally the selective pixel values for down sampled image would be written in an output text. Algorithm of the down sampling process.




Figure 2: Flow chart of the algorithm.

3 Instruction Set Architecture

In simple terms, the ISA basically defines the communication between software and hardware of the computer.

- **Data Memory (DRAM):** DRAM is used to store the pixel values of input image. With 65536 memory locations and 8 bits width.
- **Instruction Memory:** ROM with 256 memory locations and 8 bits word size.
- **Memory Address Register (MAR):** It is a 16-bit address register, used to point the addresses in DRAM.
- **Program Counter (PC):** 8-bit register, used to point the address of next instruction in Instruction Memory.
- **Memory Buffer Register Unit (MBRU):** 8-bit register, used to store the location of instructions.
- **Arithmetic and Logical Unit (ALU):** A, B are input buses and C is the output bus.
- **Control Unit:** Used to generate control signals to processor according to MBRU input.
- **Accumulator:** 24-bit accumulator with direct access to ALU.
- **Registers (X, Y, Z, CV, DCV, C1):** 16-bit General Purpose Registers, used to store intermediate values.
- **Buses (A, B, C):** 16 bit wires, used for data transmissions between DRAM, IRAM, ALU and the Registers.

3.1 Datapath




Figure 3: Data path of the processor

3.2 Instruction Set

Memory address	Instruction	Micro instruction	Breakdown of micro instruction
0	Fetch	FETCH1	MBRU \leftarrow IRAM[PC]; FETCH
1		FETCH2	PC \leftarrow PC + 1
2	NOP	NOP	No operation
3	LDAC	LDAC1	MDR \leftarrow DRAM[MAR]; READ
4		LDAC2	AC \leftarrow MDR
5	STAC	STAC1	MDR \leftarrow AC
6		STAC2	DRAM[MAR] \leftarrow MDR; WRITE
7	CLAC		AC \leftarrow 0; Z \leftarrow 1
8	MVAVMAR		MAR \leftarrow AC
9	MVACCV		CV \leftarrow AC
10	MVACC		C \leftarrow AC
11	MVACDCV		DCV \leftarrow AC
12	MVACX		X \leftarrow AC
13	MVACY		Y \leftarrow AC
14	MVACZ		Z \leftarrow AC
15	MVCV		AC \leftarrow CV
16	MVC		AC \leftarrow C
17	MVACC		C \leftarrow AC
18	MVDCV		AC \leftarrow DCV
19	MVX		AC \leftarrow X
20	MVZ		AC \leftarrow Z
21	INAC		AC \leftarrow AC+1
22	DEAC		AC \leftarrow AC-1; IF AC == 0 THEN Z = 1 ELSE Z = 0
23	ADDX		AC \leftarrow AC+X
24	ADDZ		AC \leftarrow AC+Z
25	SUBX		AC \leftarrow AC-X; IF AC == 0 THEN Z == 1 ELSE Z == 0
26	SUBY		AC \leftarrow AC+Y; IF AC == 0 THEN Z == 1 ELSE Z == 0
27	DIV		AC \leftarrow AC \gg 4
28	MUL2		AC \leftarrow AC \ll 1
29	MUL4		AC \leftarrow AC \ll 2
30	MUL256		AC \leftarrow AC \ll 8
31	JUMP	JUMP1	MBRU \leftarrow IRAM[PC]; FETCH
32		JUMP2	C \leftarrow MBRU
33		JUMP3	PC \leftarrow C
34	JMPZ	JMPZN1 (Z = 0)	PC \leftarrow PC+1
35		JMPZY1 (Z = 1)	MBRU \leftarrow IRAM[PC]; FETCH
36		JMPZY2 (Z = 1)	C \leftarrow MBRU
37		JMPZY3 (Z = 1)	PC \leftarrow C
38	JMNZ	JMNZN1 (Z = 1)	PC \leftarrow PC+1
39		JMNZY1 (Z = 0)	MBRU \leftarrow IRAM[PC]; FETCH
40		JMNZY2 (Z = 0)	C \leftarrow MBRU
41		JMNZY3 (Z = 0)	PC \leftarrow C

Figure 4: Instruction set

3.3 Control Unit

Control unit contains all control signals in Look Up table as bit patterns. In our purpose each control signal consists of 30 bits.

3.4 ALU

Here A bus and B bus are the two inputs of ALU. Here AC is directly connected through A bus. C bus and Z flag are the two outputs. Control signal for the ALU consists of 4 bits, denoting the tasks need to be performed.

ALU		
ALU command	Task	Bit pattern
NONE	None	0000'
ADD	$C=A+B$	0001'
SUB	$C=A - B$; IF $C=0$ THEN $Z=1$ ELSE $Z=0$	0010'
PASSATOC	$C=A$	0011'
PASSBTOC	$C=B$	0100'
INCAC	$C=A+1$	0101'
DECAC	$C=A - 1$; IF $C=0$ THEN $Z=1$ ELSE $Z=0$	0110'
LSHIFT1	$C=A<<1$	0111'
LSHIFT2	$C=A<<2$	1000'
LSHIFT8	$C=A<<8$	1001'
RSHIFT4	$C=A>>4$	1010'
RESET	$C=0$	1011'

Figure 5: ALU control signal breakdown

3.5 Mux for B bus

B bus would get inputs from multiple registers such as PC, MDR, MBRU, X, Y, Z, C1, CV and DCV. So there should be a multiplexer to ensure that B bus accepts only one input at a time. Therefore, read enable signal for B bus should be decided from the selection bit of the multiplexer.

B bus read enable signals	
Register	Selection bit
None	0000'
MDR	0001'
PC	0010'
MBRU	0011'
C	0100'
CV	0101'
DCV	0110'
X	0111'
Y	1000'
Z	1001'

Figure 6: Read enable signals for B bus

3.6 Assembly code of the Algorithms

The assembly code of our algorithm should be stored in Instruction ROM of the processor. So that the PC could point those instructions that need to be executed.

```
CLAC
MVACMAR
LDAC
```

```
CLAC
MVACMAR
LDAC
INAC
MVACX
MVX
INAC
INAC
MVACCV
CLAC
INAC
MVACDCV
CLAC
INAC
INAC
MVACC
```

```
MVX
```

DEAC
DEAC
MUL256
MVACY

CLAC
MVACZ
MVCV
SUBL
DEAC
MVACMAR
LDAC
ADDT MVACZ
MVCV
SUBL
MVACMAR
LDAC
MUL2
ADDT
MVAC
MVCV
SUBL
INAC
MVACMAR
DAC
ADDT
MVACZ
MVCV
DEAC
MVACMAR
LDAC
MUL2
ADDT
MVACZ
MVCV
MVACMAR
LDAC
MUL4
ADDT
MVACZ
MVCV
INAC
MVACMAR
LDAC
MUL2
ADDT
MVACZ
MVCV
ADDL
DEAC
MVACMAR
LDAC
ADDT

MVACZ
MVCV
ADDL
MVACMAR
LDAC
MUL2
ADDT
MVACZ
MVCV
ADDL
INAC
MVACMAR
LDAC
ADDT
MVACZ

MVZ
DIV
MVDCV
MVACMAR
STAC

MVDCV
INAC
MVACDCV
MVC
INAC
INAC
MVACC
MVCV
INAC
INAC
MVACCV
MVCV
SUBE
JMPZ;
L1

MVC
SUBL
JMNZ
L2
MVCV ADDL
INAC
INAC
MVACCV; CLAC
INAC
INAC
MVACC
JUMP
L2
NOP

3.7 Instruction Cycle

Instruction cycle defines the basic operating mechanism of a computer, which consists of 3 stages in it. This cycle would continue until all instructions are finished.

- **FETCH:**

Fetch retrieves required instructions from the specific location. It fetches the instruction denoted by PC from Instruction ROM and stores it in MBRU. PC would get incremented after each cycle of FETCH to point the next instruction need to be fetched.

```
FETCH1    MBRU  $\leftarrow$  IRAM[PC]; FETCH
FETCH2    PC  $\leftarrow$  PC + 1
```

- **DECODE:**

In the decoding phase, the fetched instruction in MBRU would be sent to Control Unit. Control unit would decode the instruction and generate corresponding control signals.

- **EXECUTE**

1. **NOP**

This means No Operation and is used when there is a need of waiting cycle in order to get the processed data at end point.

2. **LDAC**

This means loading AC with the data in DRAM. First the data in the location pointed by MAR would be read and loaded in MDR. Then it would be loaded to AC.

```
LDAC1     MDR  $\leftarrow$  DRAM[MAR]; READ
LDAC2     AC  $\leftarrow$  MDR
```

3. **STAC**

Here the data in AC would be loaded to MDR first and then written to the address pointed by MAR in DRAM.

```
STAC1     MDR  $\leftarrow$  AC
STAC2     DRAM[MAR]  $\leftarrow$  MDR; WRITE
```

4. **CLAC**

AC would be cleared and set to zero. Zero flag would be issued.

```
CLAC      AC  $\leftarrow$  0; Z  $\leftarrow$  1
```

5. **MVACMAR, MVACCV, MVACC, MVACDCV, MVACX, MVACY, MVACZ**

These instructions mean to move the data in AC to the specific registers.

```
MVAVMAR   MAR  $\leftarrow$  AC
MVACCV    CV  $\leftarrow$  AC
MVACC     C  $\leftarrow$  AC
MVACDCV   DCV  $\leftarrow$  AC
MVACX     X  $\leftarrow$  AC
MVACY     Y  $\leftarrow$  AC
MVACZ     Z  $\leftarrow$  AC
```

6. **MVCV, MVC, MVDCV, MVX, MVZ**

These instructions mean to move the data in specific registers to AC.

MVCV $AC \leftarrow CV$
MVC $AC \leftarrow C$
MVACC $C \leftarrow AC$
MVDCV $AC \leftarrow DCV$
MVX $AC \leftarrow X$
MVZ $AC \leftarrow Z$

7. **INAC**

Here the value in AC would be incremented by 1.

INAC $AC \leftarrow AC+1$

8. **DEAC**

The value in AC would be decremented by 1. If that value set to zero, then zero flag would be issued. ($Z = 1$)

DEAC $AC \leftarrow AC-1$; IF $AC == 0$ THEN $Z = 1$ ELSE $Z = 0$

9. **ADDX, ADDZ**

Here the value in specific register would be added with the value in AC and the result would be loaded to AC.

ADDX $AC \leftarrow AC+X$
ADDZ $AC \leftarrow AC+Z$

10. **SUBX, SUBY**

Here the value in specific register would be subtracted from the value in AC and result would be loaded to AC.

SUBX $AC \leftarrow AC-X$; IF $AC == 0$ THEN $Z == 1$ ELSE $Z == 0$.
SUBY $AC \leftarrow AC+Y$; IF $AC == 0$ THEN $Z == 1$ ELSE $Z == 0$.

11. **DIV**

Here the value in AC would be divided by 16. This is equivalent to shift the binary value to right by 4 digits.

DIV $AC \leftarrow AC \gg 4$

12. **MUL2, MUL4, MUL256**

These instructions mean to multiply the value in AC by 2, 4, 256 respectively. So it is equivalent to left by 1, 2, 8 digits respectively.

MUL2 $AC \leftarrow AC \ll 1$
MUL4 $AC \leftarrow AC \ll 2$
MUL256 $AC \leftarrow AC \ll 8$

13. **JUMP**

Here the address pointed by the current PC would be read from IRAM and loaded to MBRU. Then it would be loaded to the register C and then to the PC. So that

specific location could be fetched in the next fetch cycle.

```
JUMP1      MBRU  $\leftarrow$  IRAM[PC]; FETCH
JUMP2      C  $\leftarrow$  MBRU
JUMP3      PC  $\leftarrow$  C
```

14. **JMPZ**

This would check the Zero flag. If $Z = 0$, then the PC value would be incremented by 1. If $Z = 1$, then it would proceed steps similar to JUMP instruction.

```
JMPZN1 (Z = 0)      PC  $\leftarrow$  PC+1
JMPZY1 (Z = 1)      MBRU  $\leftarrow$  IRAM[PC]; FETCH
JMPZY2 (Z = 1)      C  $\leftarrow$  MBRU
JMPZY3 (Z = 1)      PC  $\leftarrow$  C
```

15. **JMNZ**

Here this would check zero flag. If $Z = 1$, then the PC value would be incremented by 1. If $Z = 0$, then it would proceed steps similar to JUMP instruction..

```
JMNZN1 (Z = 1)      PC  $\leftarrow$  PC+1
JMNZY1 (Z = 0)      MBRU  $\leftarrow$  IRAM[PC]; FETCH
JMNZY2 (Z = 0)      C  $\leftarrow$  MBRU
JMNZY3 (Z = 0)      PC  $\leftarrow$  C
```


3.8 State Diagram




Figure 7: State Diagram

4 RTL Modules

4.1 Processor.




Figure 8: Processor

- **Inputs:**

- **Clk** : Generated clock input.
- **enable** : To activate the processor unit.
- **din**: To input data to be processed.

- **Outputs:**

- **addr out**: The location in the memory to which the output data should be written.
- **dout**: Data to be written
- **read**: READ enable signal.
- **write**: WRITE enable signal.
- **finish**: This indicates the user that the processor has completed its task.

- **Processor consists of following modules which are vital for the down sampling.**

- ALU.
- Control store.
- PC.
- IRAM.
- MBRU.
- Decoder.
- 7 GPR.
- MDR.
- MAR.

All the registers used inside the processor can be overlooked in the following table.

Register	Size	Inputs	Outputs
GPR	24-bit	Enable, Data In	Data out
AC	24-bit	Enable, Data In	ALU
MAR	16-bit	Enable, Data In	Data memory
MDR	8-bit	Enable, Data In, Data memory	Data memory
PC	8-bit	Enable, Data In, Inc	Data out, Instruction memory
MBDU	8-bit	Enable, Data In, Instruction Memory	Data out, Control unit

Table 1: Registers

4.1.1 General Purpose Register




Figure 9: GPR

- **Purpose:** Store intermediate values of data during processing.
- **Input Strategy:** When the enabling bit (EN) is set to 1, the data available at the input to the register will be written to the register at the next proceeding positive edge of the clock.
- **Output Strategy:** A demultiplexer is placed to select which register can output data to the bus line.

4.1.2 Accumulator




Figure 10: AC

- **Purpose:** Provides a direct input to the ALU.
- Accumulator has a CLAC command to clear the current register value. This command is essential to reduce the number of clock cycles and the complexity of architecture. The AC is used in almost all the ALU operation because of the direct Data in to the ALU.

4.1.3 Memory Address Register (MAR)




Figure 11: MAR

- **Purpose:** The locations in Data Memory to which data is required to be written will be provided by MAR while the data will be available at the MDR.
- **Output Strategy:** Data out of MAR is directly connected to the B bus without a Demultiplexer.

4.1.4 Program Counter (PC)




Figure 12: PC

- **Purpose:** This stores the address of the next instruction to be executed.
- **Input Strategy:** The address is incremented by one if the INC is kept HIGH at a positive clock edge.
- **Output Strategy:** 'INC' to increment the address by one so that it is not necessary for the address to be sent to the ALU for incrementing.

4.1.5 Memory Buffer Register Unit (MBRU)




Figure 13: MBRU

- **Purpose:** This is to keep the location of an instruction to be decoded by the control Unit in order to generate relevant control signals.
- **Input Strategy:** If fetch is HIGH at the positive edge of the clock, then the location of an instruction is read and loaded from the Instruction Memory.

- **Output Strategy:** This sends an address of an instruction which is later decoded by the instruction store.

4.1.6 Instruction ROM (IRAM)




Figure 14: IRAM

- **Purpose:** Retrieve the location of the instruction which has to be executed to the MBRU once PC requests it.
- **Input:**
 - **clk:** Generated clock input.
 - **address:** The counter keeps track of the next instruction which has to be executed next.
- **Output:**
 - **d_{out}** : The address of the next instruction which has to be executed next.

4.1.7 Control Unit

- **Purpose:** This retrieves the hard wired control signals which is to be routed to each register in order to perform a certain instruction.
- **Input:**
 - **MBRU :** Fetched address of the instruction which has to be executed next.
 - **clk:** Generated clock input.
 - **Z flag:** the result from the ALU whether an operation returns zero or not.
 - **address:** Address of the next micro instruction, if the previous instruction leads to another micro instruction.
 - **enable:** TO enable the module.
- **Output:**
 - **MIR:** 30 bit register which is hardwired to required modules in order to perform an instruction.
 - **finish:** A flag to indicate the end of the operation.

4.1.8 Decoder

- **Purpose:** This module is a multiplexer to pass the outputs of the registers to the B bus one at a time.
- **Input:**




Figure 15: Decoder

- All the outputs of the registers shown in the figure.
- B bus control (selection bit)

- **Output:**

- B bus

4.1.9 Overview of the processor




Figure 16: Processor

4.2 Data RAM (DRAM)

- **Purpose:** Keeps the actual data (in our example, the image) inside it, to be processed by the processor.
- **Input:**




Figure 17: DRAM

- **addr:** Address of the memory location to which data should be written/read from provided by the user or processor.
- **clk:** Generated clock input.
- **din:** 8-bit long set of data to be written to the DRAM at the provided location.
- **read:** READ enable signal from the user or processor.
- **write:** WRITE enable signal from the user or processor.

- **Output:**

- **dout:** A byte of data read from the DRAM (May be the input to the Processor or to the user selected accordingly from a demultiplexer).

4.3 Clock Generator

The architecture of the processor is inherently depended on a clock. The designed processor must read data from the memory, process them through the ALU and write the processed data back in relevant locations within the negative edge and the positive edge of the clock. Clock with 20 ns time period is used here because a clock with every possible time period is not available in industry. Depending on that high frequency clock, we are generating a clock with a time period of 160ns.


More importantly, we implement this clock only in test bench. The reason is in FPGAs there is a clock exists as a hardware.

4.4 Multiplexers and Demultiplexers


Multiplexer means many into one. A multiplexer is a circuit used to select and route any one of the several input signals to a single output. Few types of multiplexer are 2-to-1, 4-to-1, 8-to-1 and 16-to-1 multiplexer.

In our architecture, we use multiplexers in only one size : 2 to 1 MUX.


- **address mux:** This is to route the address from the user or from the processor to the DRAM.
- **data mux:** This is to route the data from the user or from the processor to the DRAM to write.
- **read mux:** This is to route the read enable signal given by the user or by the processor to the DRAM.
- **write mux:** This is to route the write enable signal given by the user or by the processor to the DRAM.




((a)) address mux



((b)) data mux



((c)) read mux



((d)) write mux

Figure 18: Multiplexers

Demultiplexer means one to many. A demultiplexer is a circuit with one input and many outputs. By applying control signal, we can steer any input to the output. Few types of demultiplexer are 1-to-2, 1-to-4, 1-to-8 and 1-to-16 demultiplexer. Here we use only one demultiplexer.

- **data demux:** This is to route the data output from the DRAM to the user or to the processor.




Figure 19: data demux

4.5 Over view of the CPU




Figure 20: CPU

5 Results and Error analysis

A gray-scale 256 by 256 image was used for the down-sampling task. Then the downsampled image obtained from the processor was compared against a python simulated (assembly code algorithm implemented in python) image. The following metrics were used for the comparison.

- **Absolute Error**

Here, the sum of absolute element-wise difference was obtained for the each comparison.

$$AbsoluteError = \sum_{i=1}^L \sum_{j=1}^L |R_{i,j} - C_{i,j}|$$

- **Sum of Square Differences**

Here, the sum of the squares of element-wise difference was obtained for the each comparison.

$$SSD = \sum_{i=1}^L \sum_{j=1}^L |R_{i,j} - C_{i,j}|^2$$

- **Maximum Error**

Here, the maximum difference between in the corresponding pixels of the two images was obtained.






Figure 21: Original image



((a)) Python downsampled



((b)) FPGA downsampled

Figure 22: Downsampled images

- Absolute Error with simulated image : 0.0
- Sum of Square Difference with simulated image : 0.0
- Maximum Error with simulated image : 0.0

6 Appendix

6.1 Python Codes

6.1.1 Image Downsampling

```
import numpy as np
import cv2
import matplotlib.pyplot as plt

# downsample the image using python script
im = cv2.imread('dog1.jpg', cv2.IMREAD_GRAYSCALE)
im = cv2.resize(im, (256, 256))

plt.subplot(1, 2, 1)
plt.imshow(im, cmap='gray')

im = im.astype(int)
im = np.reshape(im, (65536))
image = np.zeros(65536)
image[0] = 255
image[1:65536] = im[0:65535]
currVal = 258
counter = 2
downsampledCurrVal = 1
while (currVal < 65024):
    if (counter == 256):
        currVal = currVal + 258
        counter = 2

    total = 0
    total += image[currVal - 256 - 1]
    total += image[currVal - 256] * 2
    total += image[currVal - 256 + 1]
    total += image[currVal - 1] * 2
    total += image[currVal] * 4
    total += image[currVal + 1] * 2
    total += image[currVal + 256 - 1]
    total += image[currVal + 256] * 2
    total += image[currVal + 256 + 1]
    avg = total / 16
    image[downsampledCurrVal] = avg
    downsampledCurrVal += 1
    counter += 2
    currVal += 2

downsampled_im = image[1:127*127+1]
downsampled_im = np.reshape(downsampled_im, (127, 127))

downsampled_im = np.array(downsampled_im, dtype=np.uint8)

plt.subplot(1, 2, 2)
plt.imshow(downsampled_im, cmap='gray')
```

6.1.2 Generating Input Text

```
import cv2
import numpy as np

im = cv2. imread ( 'dog1.jpg' ,cv2.IMREAD_GRAYSCALE)
im = cv2. resize (im ,(256 ,256) )

plt.subplot(1,2,1)
plt.imshow(im,cmap='gray')

im = np. reshape (im ,65536)
print(im)
a= open("input2.txt","w")
a.write("255")
for i in im[: -1]:
    a.write("\n%s"%i)
a.close()
```

6.1.3 Generating Image From Text

```
txt_file = open("D:/DownSampling/Downsampling/Downsampling.sim/sim_1/behav/xsim/
l=127
file_content = txt_file.read()
txt_file.close()
content_list = file_content.split("\n")
for i in range (1,len(content_list)-1):
    content_list[i]=int(content_list[i])

content_list.pop(0)
content_list.pop()

print(len(content_list))

for i in range (len(content_list)):
    if (type(content_list[i])!=int):
        print("i"+str(i))
        print(content_list[i])

arr=np.asarray(content_list ,dtype=np.float32)
txt_file.close()
print(type(arr[0]))
arr = np. reshape ( arr ,(1,1) )

import matplotlib.pyplot as plt
plt.imshow(arr , cmap='gray' , vmin=0, vmax=255)
```

6.1.4 Error Analysis

```
#im = cv2. imread ( 'einstein.jpg' ,cv2.IMREAD_GRAYSCALE)
#im = cv2. resize (im ,(127,127) )

arr = np. reshape ( arr ,127*127)
```

```

downsampled_im = np. reshape ( downsampled_im , 127*127)

diff = arr - downsampled_im

abs_error = sum(np. absolute ( diff ) )

sq_diff = np. power ( diff ,2)
SSD = sum( sq_diff )

non_zero_count = np. count_nonzero ( diff )
max_error = max(np. absolute ( diff ) )

print ( 'Absolute Error with simulated image : ',abs_error )
print ( 'Sum of Square Difference with simulated image : ',SSD)
print ( 'Maximum Error with simulated image :' ,max_error )

```

6.2 Verilog Codes

6.2.1 ALU

```

1  module ALU(
2
3  input  [15:0] A_bus ,
4  input  [15:0] B_bus ,
5  input  [3:0] operation ,
6  input  enable ,
7  input  clk ,
8  output reg [15:0] C_bus ,
9  output reg Z_flag
10
11 );
12
13 //Define the ALU operations
14
15 parameter ADD = 4'b0001;
16 parameter SUB = 4'b0010;
17 parameter PASSATOC = 4'b0011;
18 parameter PASSBTOC = 4'b0100;
19 parameter INCAC = 4'b0101;
20 parameter DECAC = 4'b0110;
21 parameter LSHIFT1 = 4'b0111;
22 parameter LSHIFT2 = 4'b1000;
23 parameter LSHIFT8 = 4'b1001;
24 parameter RSHIFT4 = 4'b1010;
25 parameter RESET = 4'b1011;
26
27 initial begin
28     C_bus = 16'b0;
29     Z_flag = 1'b0;
30 end
31
32 reg [1:0] state = 2'b0;
33 reg start = 1'b0;
34
35 always@(posedge enable)
36     start <= 1'b1;
37
38 always@(negedge clk)

```

```

39         begin
40             if (start) begin
41                 state = state+ 2'b01;
42             end
43         end
44
45     always@(posedge clk)
46     begin
47         if (start) begin
48             if (state == 2'b11) begin
49                 case(operation)
50                     ADD : C_bus <= A_bus + B_bus;
51
52                     SUB : begin
53                         C_bus = A_bus - B_bus;
54                         Z_flag = (C_bus == 16'b0) ? 1'b1 : 1'b0;
55                     end
56
57                     PASSATOC : C_bus <= A_bus;
58                     PASSBTOC : C_bus <= B_bus;
59                     INCAC : C_bus <= A_bus + 1;
60                     DECAC : C_bus <= A_bus - 1;
61                     LSHIFT1 : C_bus <= A_bus << 1;
62                     LSHIFT2 : C_bus <= A_bus << 2;
63                     LSHIFT8 : C_bus <= A_bus << 8;
64                     RSHIFT4 : C_bus <= A_bus >> 4;
65                     RESET : C_bus <= 16'b0;
66
67                 endcase
68             end
69         end
70     end
71 endmodule

```

6.2.2 Control Unit

```

1
2 module control_unit(
3
4     input enable,
5     input clk,
6     input Z_flag,
7     input [7:0] addr,
8     input [7:0] MBRU,
9     output reg [29:0] MIR,
10    output finish
11
12 );
13
14 reg [1:0] state = 2'b00;
15 reg start = 1'b0;
16 reg [29:0] ROM[0:42];
17 reg check = 1'b0;
18
19 assign finish = check;
20
21
22
23 parameter JMPZ1 = 8'd33;
24 parameter JMPZN1 = 8'd34;

```

```

25 parameter JMPZY1 = 8'd35;
26 parameter JMNZ1 = 8'd38;
27 parameter JMNZY1 = 8'd39;
28 parameter JMNZN1 = 8'd40;
29 parameter FETCH2 = 8'd1;
30 parameter NOP = 8'd2;
31
32 initial
33     begin
34         MIR = 30'b0;
35     end
36
37 always@(posedge enable)
38     start = 1'b1;
39
40 always@(negedge clk)
41     begin
42         if (start) begin
43             state = state+ 2'b01;
44         end
45     end
46
47 always @(posedge clk)
48     begin
49         if(enable) begin
50             if (state == 2'b11) begin
51                 case(addr)
52                     FETCH2 : MIR = {MBRU,ROM[FETCH2][21:0]};
53                     JMPZ1  : if (Z_flag == 1'b0) MIR = {8'd34,ROM[JMPZ1][21:0]};
54                             else MIR = {8'd35,ROM[JMPZ1][21:0]};
55                     JMNZ1  : if (Z_flag == 1'b1) MIR = {8'd39,ROM[JMNZ1][21:0]};
56                             else MIR = {8'd40,ROM[JMNZ1][21:0]};
57                     NOP    : check = 1'b1;
58                     default : MIR = ROM[addr];
59                 endcase
60             end
61         end
62     end
63
64 //microinstruction
65 initial
66     begin
67
68 ROM[0]   = 30'b00000001_0000_0000000000_100_0_0000; //FETCH1
69 ROM[1]   = 30'bXXXXXXXX_0000_0000000000_000_1_0000; //FETCH2
70 ROM[2]   = 30'b00000010_0000_0000000000_000_0_0000; //NOP
71 ROM[3]   = 30'b00000100_0000_0000000000_010_0_0000; //LDAC1
72 ROM[4]   = 30'b00000000_0100_0000000001_000_0_0001; //LDAC2
73 ROM[5]   = 30'b00000110_0011_0100000000_000_0_0000; //STAC1
74 ROM[6]   = 30'b00000000_0000_0000000000_001_0_0000; //STAC2
75 ROM[7]   = 30'b00000000_1011_0000000001_000_0_0000; //CLAC
76 ROM[8]   = 30'b00000000_0101_0000000001_000_0_0000; //INAC
77 ROM[9]   = 30'b00000000_0110_0000000001_000_0_0000; //DEAC
78 ROM[10]  = 30'b00000000_0001_0000000001_000_0_1000; //ADDZ
79 ROM[11]  = 30'b00000000_0001_0000000001_000_0_0100; //ADDX
80 ROM[12]  = 30'b00000000_0010_0000000001_000_0_1001; //SUBY
81 ROM[13]  = 30'b00000000_0010_0000000001_000_0_0100; //SUBX
82 ROM[14]  = 30'b00000000_1010_0000000001_000_0_0000; //DIV
83 ROM[15]  = 30'b00000000_0111_0000000001_000_0_0000; //MUL2
84 ROM[16]  = 30'b00000000_1000_0000000001_000_0_0000; //MUL4
85 ROM[17]  = 30'b00000000_1001_0000000001_000_0_0000; //MUL256

```



```

86 ROM[18] = 30'b00000000_0011_1000000000_000_0_0000; //MVACMAR
87 ROM[19] = 30'b00000000_0011_0000100000_000_0_0000; //MVACCV
88 ROM[20] = 30'b00000000_0011_0000010000_000_0_0000; //MVACC
89 ROM[21] = 30'b00000000_0011_0000001000_000_0_0000; //MVACDCV
90 ROM[22] = 30'b00000000_0011_0001000000_000_0_0000; //MVACX
91 ROM[23] = 30'b00000000_0011_0000000010_000_0_0000; //MVACY
92 ROM[24] = 30'b00000000_0011_0000000100_000_0_0000; //MVACZ
93 ROM[25] = 30'b00000000_0100_0000000001_000_0_0101; //MVCV
94 ROM[26] = 30'b00000000_0100_0000000001_000_0_0110; //MVC
95 ROM[27] = 30'b00000000_0100_0000000001_000_0_0111; //MVDCV
96 ROM[28] = 30'b00000000_0100_0000000001_000_0_1000; //MVZ
97 ROM[29] = 30'b00000000_0100_0000000001_000_0_0100; //MVX
98
99
100 ROM[30] = 30'b00011111_0000_0000000000_100_0_0000; // JUMP1
101 ROM[31] = 30'b00100000_0100_0000000000_000_0_0011; // JUMP2
102 ROM[32] = 30'b00000000_0000_0010000000_000_0_0000;
103
104 ROM[33] = 30'bxxxxxxxx_0000_0000000000_000_0_0000; // JMPZ1
105 ROM[34] = 30'b00000000_0000_0000000000_000_1_0000; // JMPZN1
106 ROM[35] = 30'b00100100_0000_0000000000_100_0_0000; // JMPZY1
107 ROM[36] = 30'b00100101_0100_0000000000_000_0_0011; // JMPZY2
108 ROM[37] = 30'b00000000_0000_0010000000_000_0_0000; // JMPZY3
109
110 ROM[38] = 30'bxxxxxxxx_0000_0000000000_000_0_0000; // JMNZ1
111 ROM[39] = 30'b00000000_0000_0000000000_000_1_0000; // JMNZY1
112 ROM[40] = 30'b00101001_0000_0000000000_100_0_0000; // JMNZN1
113 ROM[41] = 30'b00101010_0100_0000000000_000_0_0011; // JMNZN2
114 ROM[42] = 30'b00000000_0000_0010000000_000_0_0000; // JMNZN3
115
116
117
118
119 end
120
121 endmodule

```

6.2.3 IRAM

```

1
2
3
4
5
6
7 module IRAM(
8
9     input clk,
10    input [7:0] addr,
11    output reg [7:0] dout
12
13 );
14
15 reg [7:0] ROM [120:0];
16
17 // Assembly Instructions
18
19 parameter FETCH = 8'd0;
20 parameter NOP = 8'd2;
21 parameter LDAC = 8'd3;

```

```

22 parameter STAC = 8'd5;
23 parameter CLAC = 8'd7;
24 parameter INAC = 8'd8;
25 parameter DEAC = 8'd9;
26 parameter ADDT = 8'd10;
27 parameter ADDL = 8'd11;
28 parameter SUBE = 8'd12;
29 parameter SUBL = 8'd13;
30 parameter DIV = 8'd14;
31 parameter MUL2 = 8'd15;
32 parameter MUL4 = 8'd16;
33 parameter MUL256 = 8'd17;
34 parameter MVACMAR = 8'd18;
35 parameter MVACCV = 8'd19;
36 parameter MVACC = 8'd20;
37 parameter MVACDCV = 8'd21;
38 parameter MVACX = 8'd22;
39 parameter MVACY = 8'd23;
40 parameter MVACZ = 8'd24;
41 parameter MCV = 8'd25;
42 parameter MVC = 8'd26;
43 parameter MVDCV = 8'd27;
44 parameter MVZ = 8'd28;
45 parameter MVX = 8'd29;
46 parameter JUMP = 8'd30;
47 parameter JMPZ = 8'd33;
48 parameter JMNZ = 8'd38;
49 parameter L1 = 8'd120;
50 parameter L2 = 8'd21;
51
52
53 always @(posedge clk)
54     begin
55         dout <= ROM[addr];
56     end
57
58 initial
59     begin
60
61     ROM[0] = CLAC;
62     ROM[1] = MVACMAR;
63     ROM[2] = LDAC;
64     ROM[3] = INAC;
65     ROM[4] = MVACX;
66     ROM[5] = MVX;
67     ROM[6] = INAC;
68     ROM[7] = INAC;
69     ROM[8] = MVACCV;
70     ROM[9] = CLAC;
71     ROM[10] = INAC;
72     ROM[11] = MVACDCV;
73     ROM[12] = CLAC;
74     ROM[13] = INAC;
75     ROM[14] = INAC;
76     ROM[15] = MVACC;
77
78     ROM[16] = MVX;
79     ROM[17] = DEAC;
80     ROM[18] = DEAC;
81     ROM[19] = MUL256;
82     ROM[20] = MVACY;

```

```

83
84 ROM[21] = CLAC ;
85 ROM[22] = MVACZ ;
86 ROM[23] = MVCV ;
87 ROM[24] = SUBL ;
88 ROM[25] = DEAC ;
89 ROM[26] = MVACMAR ;
90 ROM[27] = LDAC ;
91 ROM[28] = ADDT ;
92 ROM[29] = MVACZ ;
93 ROM[30] = MVCV ;
94 ROM[31] = SUBL ;
95 ROM[32] = MVACMAR ;
96 ROM[33] = LDAC ;
97 ROM[34] = MUL2 ;
98 ROM[35] = ADDT ;
99 ROM[36] = MVACZ ;
100 ROM[37] = MVCV ;
101 ROM[38] = SUBL ;
102 ROM[39] = INAC ;
103 ROM[40] = MVACMAR ;
104 ROM[41] = LDAC ;
105 ROM[42] = ADDT ;
106 ROM[43] = MVACZ ;
107 ROM[44] = MVCV ;
108 ROM[45] = DEAC ;
109 ROM[46] = MVACMAR ;
110 ROM[47] = LDAC ;
111 ROM[48] = MUL2 ;
112 ROM[49] = ADDT ;
113 ROM[50] = MVACZ ;
114 ROM[51] = MVCV ;
115 ROM[52] = MVACMAR ;
116 ROM[53] = LDAC ;
117 ROM[54] = MUL4 ;
118 ROM[55] = ADDT ;
119 ROM[56] = MVACZ ;
120 ROM[57] = MVCV ;
121 ROM[58] = INAC ;
122 ROM[59] = MVACMAR ;
123 ROM[60] = LDAC ;
124 ROM[61] = MUL2 ;
125 ROM[62] = ADDT ;
126 ROM[63] = MVACZ ;
127 ROM[64] = MVCV ;
128 ROM[65] = ADDL ;
129 ROM[66] = DEAC ;
130 ROM[67] = MVACMAR ;
131 ROM[68] = LDAC ;
132 ROM[69] = ADDT ;
133 ROM[70] = MVACZ ;
134 ROM[71] = MVCV ;
135 ROM[72] = ADDL ;
136 ROM[73] = MVACMAR ;
137 ROM[74] = LDAC ;
138 ROM[75] = MUL2 ;
139 ROM[76] = ADDT ;
140 ROM[77] = MVACZ ;
141 ROM[78] = MVCV ;
142 ROM[79] = ADDL ;
143 ROM[80] = INAC ;

```

```

144 ROM[81] = MVACMAR;
145 ROM[82] = LDAC;
146 ROM[83] = ADDT;
147 ROM[84] = MVACZ;
148
149 ROM[85] = MVZ;
150 ROM[86] = DIV;
151 ROM[87] = MVDCV;
152 ROM[88] = MVACMAR;
153 ROM[89] = STAC;
154
155 ROM[90] = MVDCV;
156 ROM[91] = INAC;
157 ROM[92] = MVACDCV;
158 ROM[93] = MVC;
159 ROM[94] = INAC;
160 ROM[95] = INAC;
161 ROM[96] = MVACC;
162 ROM[97] = MVCV;
163 ROM[98] = INAC;
164 ROM[99] = INAC;
165 ROM[100] = MVACCV;
166
167 ROM[101] = MVCV;
168 ROM[102] = SUBE;
169 ROM[103] = JMPZ;
170 ROM[104] = L1;
171
172 ROM[105] = MVC;
173 ROM[106] = SUBL;
174 ROM[107] = JMNZ;
175 ROM[108] = L2;
176 ROM[109] = MVCV;
177 ROM[110] = ADDL;
178 ROM[111] = INAC;
179 ROM[112] = INAC;
180 ROM[113] = MVACCV;
181 ROM[114] = CLAC;
182 ROM[115] = INAC;
183 ROM[116] = INAC;
184 ROM[117] = MVACC;
185 ROM[118] = JUMP;
186 ROM[119] = L2;
187
188 ROM[120] = NOP;
189
190
191 end
192 endmodule

```

6.2.4 Clock Generator

```

1
2 module clock_gen(
3
4   input clk_in,
5   output clk_out
6
7 );
8

```

```

9  parameter factor = 4;
10
11  reg [7:0] counter = 0;
12  reg out = 0;
13
14  assign clk_out = out;
15
16  always @(posedge clk_in)
17      begin
18          if (counter < factor - 1) begin
19              counter <= counter + 1;
20          end
21          else begin
22              out <= ~out;
23              counter <= 0;
24          end
25      end
26
27  endmodule

```

6.2.5 Decoder

```

1  module decoder(
2
3  input  [15:0] X,CV,C,DCV,Z,Y,
4  input  [7:0] PC,
5  input  [7:0] MDR,
6  input  [7:0] MBRU,
7  input  [3:0] B_bus_ctrl,
8  output reg [15:0] B_bus
9
10 );
11
12 always@( B_bus_ctrl or X or CV or C or DCV or Z or Y or PC or MBRU or MDR)
13     begin
14         case(B_bus_ctrl)
15
16             4'b0001 : B_bus <= {16'b0,MDR};
17             4'b0010 : B_bus <= {16'b0,PC};
18             4'b0011 : B_bus <= {16'b0, MBRU};
19             4'b0100 : B_bus <= X;
20             4'b0101 : B_bus <= CV;
21             4'b0110 : B_bus <= C;
22             4'b0111 : B_bus <= DCV;
23             4'b1000 : B_bus <= Z;
24             4'b1001 : B_bus <= Y;
25             default : B_bus <= 16'b0;
26
27         endcase
28     end
29
30 endmodule

```

6.2.6 Demultiplexer

```

1
2  module demux12_8bit(
3
4  input  [7:0] data_input,

```

```

5  input selection,
6  output reg [7:0] A,
7  output reg [7:0] B
8
9  );
10
11 always @(data_input or selection)
12
13     begin
14         case(selection)
15             1'b0 : A = data_input;
16             1'b1 : B = data_input;
17         endcase
18     end
19
20 endmodule

```

6.2.7 GPR

```

1
2 module GPR(
3
4     input clk,
5     input load,
6     input [15:0] C_bus, //connects to the MIR load signal
7     output reg [15:0] data_out //connects to the B_bus (MUX)
8
9 );
10
11 always@(posedge clk)
12     begin
13         if (load) data_out <= C_bus;
14     end
15
16 endmodule

```

6.2.8 MAR

```

1 module MAR(
2
3     input clk,
4     input load,
5     input [15:0] C_bus,
6     output reg [15:0] data_address
7
8 );
9
10 always@(posedge clk)
11     begin
12         if (load) data_address <= C_bus;
13     end
14
15 endmodule

```

6.2.9 MBRU

```

1
2

```

```

3
4 module MBRU(
5
6 input clk,
7 input fetch,
8 input [7:0] ins_in,
9 output reg [7:0] ins_out
10
11 );
12
13 always@(posedge clk)
14     begin
15         if (fetch) ins_out <= ins_in;
16     end
17
18 endmodule

```

6.2.10 MDR

```

1
2
3
4 module MDR(
5
6 input clk,
7 input load,
8 input read,
9 input write,
10 input [7:0] C_bus,
11 input [7:0] data_in_DRAM,
12 output reg [7:0] data_out_Bbus,
13 output reg [7:0] data_out_DRAM
14
15 );
16
17 always@(posedge clk)
18     begin
19         if (load) data_out_Bbus <= C_bus;
20
21         if (read) data_out_Bbus <= data_in_DRAM;
22
23         if (write) data_out_DRAM <= C_bus;
24
25     end
26
27 endmodule

```

6.2.11 Multiplexer (1-bit)

```

1
2 module signal_mux(
3
4 input A,
5 input B,
6 input selection,
7 output reg data_output
8
9 );
10

```

```

11 always @(A or B or selection)
12
13     begin
14         case(selection)
15             1'b0 : data_output = A;
16             1'b1 : data_output = B;
17         endcase
18     end
19
20 endmodule

```

6.2.12 Multiplexer (8-bit)

```

1
2 module mux21_8bit(
3
4     input [7:0] A,
5     input [7:0] B,
6     input selection,
7     output reg [7:0] data_output
8
9 );
10
11 always @(A or B or selection)
12     begin
13         case(selection)
14             1'b0 : data_output = A;
15             1'b1 : data_output = B;
16         endcase
17     end
18
19 endmodule

```

6.2.13 Multiplexer (16-bit)

```

1 module mux21_16bit(
2
3     input [15:0] A,
4     input [15:0] B,
5     input selection,
6     output reg [15:0] data_output
7
8 );
9
10 always @(A or B or selection)
11     begin
12         case(selection)
13             1'b0 : data_output = A;
14             1'b1 : data_output = B;
15         endcase
16     end
17
18 endmodule

```

6.2.14 PC

```

1 module PC(
2

```



```

3  input enable,
4  input clk,
5  input finish,
6  input load, //connects to PC Write signal (for tb 0)
7  input inc, //inc signal
8  input [7:0] C_bus,
9  output reg [7:0] ins_address //MBRU ins_in
10
11 );
12 reg start = 1'b0;
13 reg [1:0] state = 2'b0;
14
15 initial begin
16     ins_address <= 8'b0;
17
18 end
19
20 always@(posedge enable)
21 start <= 1'b1;
22
23 always@(negedge clk)
24     begin
25         if (start) begin
26             state = state+ 2'b01;
27         end
28     end
29
30 always@(posedge clk)
31     begin
32         if (finish) begin ins_address <= ins_address; end
33         else if (enable) begin
34             if (load && state == 2'b11) begin
35                 ins_address <= C_bus;
36             end
37             else if (inc && state == 2'b11) begin
38                 ins_address <= ins_address + 8'b00000001;
39             end
40             else begin
41                 ins_address <= ins_address;
42             end
43         end
44     end
45
46 endmodule

```

6.2.15 Processor

```

1
2 module Processor(
3
4  input enable,
5  input clk,
6  input [7:0] din,
7  output [15:0] addr_out,
8  output [7:0] dout,
9  output finish,
10 output read,
11 output write
12
13 );

```

```

14
15 wire [7:0] ins_address;
16 wire [7:0] ins_in;
17 wire [7:0] ins_out;
18 wire [29:0] MIR;
19 wire [15:0] A_bus;
20 wire [15:0] B_bus;
21 wire [15:0] C_bus;
22 wire Z_flag;
23 wire [15:0] X_bus;
24 wire [15:0] Y_bus;
25 wire [15:0] Z_bus;
26 wire [15:0] CV_bus;
27 wire [15:0] Counter_bus;
28 wire [15:0] DCV_bus;
29 wire [7:0] MDR_bus;
30 wire finish_flag;
31
32 assign read = MIR[6];
33 assign write = MIR[5];
34
35 assign finish = finish_flag;
36
37 //Define the Control Store
38
39 control_unit CTRL_UNIT(
40 .enable(enable),
41 .clk(clk),
42 .Z_flag(Z_flag),
43 .addr(MIR[29:22]),
44 .MBRU(ins_out),
45 .MIR(MIR),
46 .finish(finish_flag)
47 );
48
49
50
51 PC PC(
52 .enable(enable),
53 .clk(clk),
54 .inc(MIR[4]),
55 .C_bus(C_bus),
56 .finish(finish_flag),
57 .ins_address(ins_address),
58 .load(MIR[15])
59 );
60
61
62 IRAM IRAM(
63 .clk(clk),
64 .addr(ins_address),
65 .dout(ins_in)
66 );
67
68
69 MBRU MBRU(
70 .clk(clk),
71 .fetch(MIR[7]),
72 .ins_in(ins_in),
73 .ins_out(ins_out)
74 );

```

```

75
76
77 GPR X(
78   .clk(clk),
79   .load(MIR[14]),
80   .C_bus(C_bus),
81   .data_out(X_bus)
82 );
83
84
85
86 GPR CV( //CURRENT VALUE
87   .clk(clk),
88   .load(MIR[13]),
89   .C_bus(C_bus),
90   .data_out(CV_bus)
91 );
92
93
94
95 GPR C( //COUNTER
96   .clk(clk),
97   .load(MIR[12]),
98   .C_bus(C_bus),
99   .data_out(Counter_bus)
100 );
101
102
103
104 GPR DCV(//DOWNSAMPLED CURRENT VALUE
105   .clk(clk),
106   .load(MIR[11]),
107   .C_bus(C_bus),
108   .data_out(DCV_bus)
109 );
110
111
112
113 GPR Z(
114   .clk(clk),
115   .load(MIR[10]),
116   .C_bus(C_bus),
117   .data_out(Z_bus)
118 );
119
120
121
122 GPR Y(
123   .clk(clk),
124   .load(MIR[9]),
125   .C_bus(C_bus),
126   .data_out(Y_bus)
127 );
128
129
130
131 GPR AC(
132   .clk(clk),
133   .load(MIR[8]),
134   .C_bus(C_bus),
135   .data_out(A_bus)

```

```

136 );
137
138
139
140 MDR MDR(
141   .clk(clk),
142   .load(MIR[16]),
143   .read(MIR[6]),
144   .write(MIR[5]),
145   .C_bus(C_bus[7:0]),
146   .data_in_DRAM(din),
147   .data_out_DRAM(dout),
148   .data_out_Bbus(MDR_bus)
149 );
150
151
152 MAR MAR(
153   .clk(clk),
154   .load(MIR[17]),
155   .C_bus(C_bus[15:0]),
156   .data_address(addr_out)
157 );
158
159
160 ALU ALU(
161   .A_bus(A_bus),
162   .B_bus(B_bus),
163   .operation(MIR[21:18]),
164   .C_bus(C_bus),
165   .enable(enable),
166   .clk(clk),
167   .Z_flag(Z_flag)
168 );
169
170
171 decoder decoder(
172   .X(X_bus),
173   .CV(CV_bus),
174   .C(Counter_bus),
175   .DCV(DCV_bus),
176   .Z(Z_bus),
177   .Y(Y_bus),
178   .PC(ins_address),
179   .MDR(MDR_bus),
180   .MBRU(ins_out),
181   .B_bus_ctrl(MIR[3:0]),
182   .B_bus(B_bus)
183 );
184
185 endmodule

```

6.2.16 Downsampling CPU

```

1
2 `timescale 1ns / 1ps
3
4 module downsampling_CPU(
5
6   input clka,
7   input enable,

```

```

8  input  [7:0] data_in,
9  input  [15:0] add_in,
10 input  data_write,
11 input  data_read,
12 input  selection,
13 output finish,
14 output [7:0] data_out
15 );
16
17 //define busses
18
19 wire [15:0] address_bus;
20 wire [7:0] data_in_bus;
21 wire [7:0] data_out_bus;
22 wire [7:0] dram_in_bus;
23 wire [15:0] dram_addr;
24 wire [7:0] dram_out;
25 wire write;
26 wire read;
27 wire dram_write;
28 wire dram_read;
29
30
31 Processor Processor(
32
33   .din(data_out_bus),
34   .enable(enable),
35   .clk(clka),
36   .dout(data_in_bus),
37   .addr_out(address_bus),
38   .finish(finish),
39   .write(write),
40   .read(read)
41 );
42
43
44 DRAM DRAM(
45
46   .addr(dram_addr),
47   .din(dram_in_bus),
48   .read(dram_read),
49   .write(dram_write),
50   .clk(clka),
51   .dout(dram_out)
52 );
53
54
55
56
57
58
59
60 mux21_8bit data_mux(
61   .B(data_in_bus),
62   .A(data_in),
63   .selection(selection),
64   .data_output(dram_in_bus)
65 );
66
67
68 signal_mux write_mux(

```

```

69 .A(data_write),
70 .B(write),
71 .selection(selection),
72 .data_output(dram_write)
73 );
74
75 signal_mux read_mux(
76 .A(data_read),
77 .B(read),
78 .selection(selection),
79 .data_output(dram_read)
80 );
81
82 mux21_16bit address_mux(
83 .B(address_bus),
84 .A(add_in),
85 .selection(selection),
86 .data_output(dram_addr)
87
88 );
89
90 demux12_8bit data_demux (
91 .data_input(dram_out),
92 .selection(selection),
93 .A(data_out),
94 .B(data_out_bus)
95
96 );
97
98
99 endmodule

```